Maple-assisted derivation of recurrence for A245868

Robert Israel

13 May 2020

We enumerate $[0\,..\,7]^2$ as follows:

```
> Pairs:= [seq(seq([i,j],j=0..7),i=0..7)]:
```

Let $b(n)$ be the column vector of 64 whose k'th entry.is the number of arrays counted by $a(n)$ whose last two entries are $Pairs(k)$.

We have the following transition matrix whose (j, k) entry is 1 if the first entry of $Pairs(j)$ is the second entry of $Pairs(i)$ and some two of the three terms $Pairs(i)_1$, $Pairs(i)_2$, $Pairs(j)_2$ sum to 7.

```
> T:= Matrix(64,64,(i,j) -> `if`(Pairs[j][1]=Pairs[i][2] and member
  (7, {Pairs[i][1]+Pairs[i][2],
    Pairs[i][1]+Pairs[j][2], Pairs[i][2]+Pairs[j][2]}), 1, 0)):
```

Then $b(n+1) = T\,b(n)$, with $b(0)$ the vector e of all 1's. We then should have $a(n) = e^T\,T^n\,e$ for all $n \geq 1$. To check this, we will compute the first few terms. For future use, it will be convenient to pre-compute some $T^n e$, and gather them together as columns of a matrix.

```
> e:= Vector(64,1):
```

```
> Te[0]:= e: L:= e:
  for nn from 1 to 22 do Te[nn]:= T . Te[nn-1]; L:= <L|Te[nn]> od:
  seq(e^%T . Te[n],n=1..22);
```

168, 712, 2368, 8840, 31176, 113024, 404264, 1455496, 5223552, 18775816, 67437448, 242306240, 870461352, 3127322696, 11235107264, 40363689352, 145010699592, 520968428032, 1871637364264, 6724074597128, 24157004951808, 86786820122120 **(1)**

The recurrence will produce a linear dependence among $T^n e$. This will show up as L having less than full column rank..

```
> for m from 1 do
    if LinearAlgebra:-Rank(L[..,1..m]) < m then printf("Success at
  m=%d\n",m); break fi
  od:
Success at m=4
```

The recurrence can then be found from the null space of the first 4 columns of L.

```
> P:= LinearAlgebra:-NullSpace(L[..,1..4])[1]:
  recurrence:= sort(a(n)= solve(add(P[i]*a(n+i-4),i=1..4),a(n)),
  [seq(a(n-i),i=0..3)]);
```

$$recurrence := a(n) = 2\,a(n-1) + 6\,a(n-2) - a(n-3) \quad \textbf{(2)}$$